[Google Pixel ROMs, Kernels, Recoveries, & Other De] me-69:
Thanks a lot for this great work!

After I f'ed up the slots several times, I wrote a small batch for a clean install including TWRP and Magisk. As it calls the flash-all.bat for installing the stock ROM, at least the 'exit' command at the end has to be deleted there. All files -- batch, stock ROM, custom ROM, TWRP, Magisk -- have to be in one directory.

@ECHO OFF

REM This batch file requires the files named below as well as the stock rom files in the same directory.
REM You'll have to modify the flash_all.bat from the stock rom by deleting the last three lines (echo, wait, exit)!
REM Please save the modified file as flash_all_mod.bat

REM Enter your filenames here, avoid spaces around the equal sign!
SET TWRP-IMG=twrp-3.2.3-1-sailfish.img
SET TWRP-ZIP=twrp-pixel-installer-sailfish-3.2.3-1.zip
SET ROM=PixelDust-X-caf-sailfish-20200707-1732.zip
SET MAGISK=Magisk-v20.4.zip

if not exist "%TWRP-IMG%" (
echo %TWRP-IMG% missing.
goto :EOF
)

if not exist "%TWRP-ZIP%" (
echo %TWRP-ZIP% missing.
goto :EOF
)

if not exist "%ROM%" (
echo %ROM% missing.
goto :EOF
)

if not exist "%MAGISK%" (
echo %MAGISK% missing.
goto :EOF
)

if not exist flash-all.bat (
echo flash-all-mod.bat missing. Is there any stock ROM in this directory? Have you modified the flash-all.bat file?
goto :EOF
)

echo.
echo Clean install ROM, TWRP and Magisk. THIS WILL ERASE YOUR DATA!
echo Please manually boot to bootloader first (power, vol-), then press any key to start stock rom installation to slot A.
pause >nul

:INSTALL_STOCK

fastboot --set-active=a
fastboot reboot bootloader
ping -n 5 127.0.0.1 >nul
call flash-all-mod.bat
echo.
echo Booting from A, please wait until 'Hi there'.
echo Then please manually boot to bootloader again and press any key to start stock rom installation to slot B.
pause >nul

fastboot --set-active=b
fastboot reboot bootloader
ping -n 5 127.0.0.1 >nul
call flash-all-mod.bat
echo.
echo Booting from B, please wait until 'Hi there'.
echo Then please manually boot to bootloader again and press any key to start TWRP from RAM.
pause >nul

:WIPE

fastboot boot "%TWRP-IMG%"
echo.
echo TWRP started with active slot B, system may be kept read only. Please wipe Dalvic, System and Data.
echo Then press any key to restart TWRP with active slot A.
pause >nul

:INSTALL_ROM_TWRP

adb reboot bootloader
ping -n 5 127.0.0.1 >nul
fastboot --set-active=a
fastboot boot "%TWRP-IMG%"
echo.
echo TWRP started with active slot A, system may be kept read only.
echo Press any key to continue copying the ROM installation file.
pause >nul

echo Copying %ROM%
adb push "%ROM%" /sdcard/
echo Copying %TWRP-ZIP%
adb push "%TWRP-ZIP%" /sdcard/
echo.
echo Now please install %ROM% and %TWRP-ZIP%.
echo After installation completed, press any key to boot the installed ROM from slot B.
pause >nul

adb reboot bootloader
ping -n 5 127.0.0.1 >nul
fastboot --set-active=b
fastboot reboot
echo.
echo Booting custom ROM from B, please complete the installation.
echo After that, press any key to begin with installing Magisk.
pause >nul

:INSTALL_MAGISK

REM adb reboot bootloader
REM ping -n 10 127.0.0.1 >nul
REM fastboot boot "%TWRP-IMG%"
REM unnecessary after TWRP is installed

adb reboot recovery
echo.
echo TWRP started with active slot B, system may be kept read only.
echo Press any key to continue copying the Magisk installation file.
pause >nul

echo Copying %MAGISK%
adb push "%MAGISK%" /sdcard/
echo.
echo Now please install %MAGISK% and then wipe cache/dalvik.
echo After that, press any key to clean up and boot again.
pause >nul

:CLEANUP

adb shell rm /sdcard/"%MAGISK%"
adb shell rm /sdcard/"%TWRP-ZIP%"
adb shell rm /sdcard/"%ROM%"

adb reboot

Feature request: Options to (un)swap navigation buttons and remove Google search and date & weather display on the home screen.